10 PREDATOR-PREY DYNAMICS

Objectives

- Set up a spreadsheet model of interacting predator and prey populations.
- Modify the model to include an explicit carrying capacity for the prey population, independent of the effect of predation.
- Explore the effects of different prey reproductive rates on the dynamics of both models.
- Explore the effects of different predator attack rates and reproductive efficiencies on the dynamics of both models.
- Evaluate the stability of these models.
- Evaluate these models in comparison to real predator and prey populations.

Suggested Preliminary Exercises: Geometric and Exponential Population Models; Logistic Population Models

INTRODUCTION

In this exercise, you will set up a spreadsheet model of interacting predator and prey populations. You will begin with the classic Lotka-Volterra predator-prey model (Rosenzweig and MacArthur 1963), which treats each population as if it were growing exponentially. After exploring the predictions of this model, you will modify it to include refuges for the prey and see how this changes the behavior of the model.

Next, you will modify the model of the prey population to include an explicit carrying capacity. This reflects the idea that the prey population may be limited by available resources in addition to any limitation by the effects of predation.

Finally, you may modify the predator model to include an explicit carrying capacity. This would represent some limitation on the predator population other than the availability of prey. Such limitation might arise from other required resources or from direct interference among predators.

Model Development

This exercise departs somewhat from the format of others in this book, because we want to follow the progression of increasingly complex and realistic models outlined above. You will build the simplest model first, make some graphs, and answer some questions about the model and its ecological meaning. Then you will return to the spreadsheet to modify the model, reexamine the same questions, and repeat this process a third time.

In the models that follow, we will use the symbols explained in Table 1.

TABLE 1 Symbols used in predator-prey models

Symbol	Name	Description
C_t	Predator population	Think "Consumer"
V_t	Prey population	Think "Victim"
R	Prey population growth	Per capita growth rate of prey population
K_c	Predator carrying capacity	Maximum sustainable predator population
K_v	Prey carrying capacity	Maximum sustainable prey population
q	Predator starvation rate	Per capita rate of mortality of predators due to starvation
a	Attack rate	The ability of a predator to find and consume prey
f	Conversion efficiency	The efficiency with which a predator converts consumed prey into predator offspring

First Model: A Classical Lotka-Volterra Predator-Prey Model

To begin, we will build a discrete-time version of the continuous-time model developed by Alfred Lotka and Vito Volterra. In this model, neither prey population nor predator population has an explicit carrying capacity. Be aware, however, that either or both may have an implicit carrying capacity imposed by the interaction between the two populations.

To model the prey population, we begin with a basic geometric model for the prey population

$$V_{t+1} = V_t + RV_t$$

and subtract the number of prey individuals killed by predators in the interval from t to $t + 1$. This number killed will depend on the number of predators: the more predators, the more prey they will kill. It will also depend on the number of prey available: the more prey, the more successful the predators. Finally, it will depend on the **attack rate**: the ability of a predator to find and consume prey. The number of prey killed in one time interval will be the product of these, or using the symbols given above, aC_tV_t. The equation for the prey population thus becomes

$$V_{t+1} = V_t + RV_t - aC_tV_t \qquad \text{Equation 1}$$

In words, the prey population grows according to its per capita growth rate minus losses to predators. Losses are determined by attack rate, predator population, and prey population.

To model the predator population, we also begin with an exponential model, in concept. However, there is a wrinkle in this model, because we cannot assume a constant per capita rate of population growth. There is no simple R for the predator population because its growth rate will depend on how many prey are caught. As in the prey model, the number of prey caught will be aC_tV_t. The growth of the predator population will depend on this number, and on the efficiency with which predators convert consumed prey into predator offspring. We will represent this conversion efficiency with the parameter f, so the per capita population growth of predators will be afV_tC_t. We should reduce this predator population growth by some quantity to represent the starvation rate of predators who fail to consume prey. This will be the product of the per capita starvation rate times the predator population: qC_t. Taking all this into account, we can write an equation for the predator population:

$$C_{t+1} = C_t + afV_tC_t - qC_t \qquad \text{Equation 2}$$

In words, the predator population grows according to the attack rate, conversion efficiency, and prey population, minus losses to starvation. Note that the product afV_t acts as the predator's R.

Having created these models, we can ask several questions about the interaction they portray, such as

- Under what conditions (i.e., parameter values) will the predator population drive the prey to extinction?
- Under what conditions will the predator population die off, leaving the prey population to expand unhindered?
- Under what conditions will predator and prey populations both persist indefinitely? What will be their population dynamics while they coexist? In other words, will one or both populations stabilize, or will they continue to change over time?

Equilibrium Solutions

As we did in the Interspecific Competition exercise, we will begin to answer these questions by seeking equilibrium solutions to Equations 1 and 2. For the prey population, we want to find values of predator and prey population sizes at which the prey population remains stable. In other words, we want to solve for $\Delta V_t = 0$.

Beginning with Equation 1

$$V_{t+1} = V_t + RV_t - aC_tV_t$$

we subtract V_t from both sides, and get

$$V_{t+1} - V_t = RV_t - aC_tV_t$$

Because $V_{t+1} - V_t = \Delta V_t$ we can substitute into the equation and get

$$\Delta V_t = RV_t - aC_tV_t$$

We are looking for a solution when $\Delta V_t = 0$, so we substitute again:

$$0 = RV_t - aC_tV_t$$

Adding aC_tV_t to both sides gives us

$$aC_tV_t = RV_t$$

Dividing both sides by V_t, we get

$$aC_t = R$$

Dividing both sides by a gives us our solution:

$$C_t = R/a \qquad \text{Equation 3}$$

In words, the prey population reaches equilibrium when the predator population equals the prey's per capita growth rate divided by the predator's attack rate. Note that this is a constant. Strangely, the equilibrium size of the prey population is not determined by this solution, which says, in effect, that the prey population can be stable at any size as long as the predator population is at the specified size.

For the predator population, we follow the same strategy, and solve for $\Delta C = 0$. Beginning with Equation 2,

$$C_{t+1} = C_t + afV_tC_t - qC_t$$

we subtract C_t from both sides, and get

$$C_{t+1} - C_t = afV_tC_t - qC_t$$

Because $C_{t+1} - C_t = \Delta C_t$, we can substitute into the equation and get

$$\Delta C_t = afV_tC_t - qC_t$$

We are looking for a solution when $\Delta C_t = 0$, so we substitute again:

$$0 = afV_tC_t - qC_t$$

Adding qC_t to both sides gives us

$$qC_t = afV_tC_t$$

Dividing both sides by C_t, we get

$$q = afV_t$$

Dividing both sides by af gives us our solution:

$$q/af = V_t \qquad \text{Equation 4}$$

In words, the predator population reaches equilibrium when the prey population equals the predator's starvation rate over the product of attack rate times conversion efficiency. Note that this is also a constant, and like the solution for the prey population, it does not specify the equilibrium size of the predator population, only the size of the prey population at which the predators are at equilibrium.

As we did in the model of interspecific competition, we can plot the population sizes of the two interacting populations on the two axes of a graph (Figure 1). The equilibrium solutions (Equations 3 and 4) then become straight-line zero net growth isoclines



Figure 1 Graph of prey and predator zero net growth isoclines (ZNGIs), according to the Lotka-Volterra model of predator-prey dynamics. The horizontal line is the ZNGI for the prey population, and horizontal arrows show areas of population increase or decrease for the prey population. The vertical line is the ZNGI for the predator population, and vertical arrows show areas of increase or decrease for the predator population.

(ZNGIs), as they did in the interspecific competition model. On this graph, the ZNGI for the prey population is a horizontal line at $C_t = R/a$ (the solid line in Figure 1), below which the prey population increases, and above which it decreases (solid arrows). The ZNGI for the predator population is a vertical line at $V_t = q/af$ (dashed line), to the left of which the predator population decreases, and to the right of which it increases (dashed arrows). Where the two lines cross—at the point $[(q/af), (R/a)]$—the two populations are at equilibrium. As in the Interspecific Competition exercise, the two populations are represented by a point on this phase diagram, and that point will trace out a trajectory through phase space as the populations change in size.

As discussed in most ecology texts, the continuous-time Lotka-Volterra model predicts that the point representing the two populations will cycle endlessly around the point where the two ZNGIs cross. The discrete-time model, however, behaves rather differently, as you will discover.

PROCEDURES

We will use the spreadsheet to explore the behavior of the model developed so far before we introduce the models with explicit prey and predator carrying capacities.

As always, save your work frequently to disk.

INSTRUCTIONS | ANNOTATION

Part 1. Discrete-Time Version of the Lotka-Volterra Model

A. Set up the spreadsheet.

1. Open a new spreadsheet and set up titles and column headings as shown in Figure 2.

Enter only the text items for now. These are all literals, so just select the appropriate cells and type them in. Note that cells B12 through C13 must be empty.

	A	B	C	D	E	F	G	H
1	***Predator-Prey Dynamics***							
2	Uses an exponentially-growing prey population, with an additional term for losses to predators.							
3	Uses an exponentially-growing predator population with per capita pop growth rate determined							
4		by prey capture and conversion efficiency.						
5								
6		**Zero net growth isoclines**			**Prey parameters**		**Predator parameters**	
7		3649.232	25.000		*R*	0.250	Starvation rate (*q*)	0.100
8		0.000	25.000				Conversion efficiency (*f*)	0.008
9		0.000	0.000				Attack rate (*a*)	0.010
10		1250.000	0.000					
11		1250.000	41.999					
12								
13	Time							
14	0	1000.000	20.000					
15	1	1050.000	19.600					
16	2	1106.700	19.286					

Figure 2

2. Set up a linear series from 0 to 100 in column A (cells A14–A114).

Enter the value 0 in cell A14.
Enter the formula **=A14+1** in cell A15. Copy this formula down to cell A114.

3. Enter the values shown for the parameters R, q, f, and a.

Type the values shown into cells F7, H7, H8, and H9.
Cells F8 and H10 remain empty for now.

4. Enter the initial population sizes (V_0 and C_0).

Enter the value 1000 into cell B14.
Enter the value 20 into cell C14.
Leave cells B12 through C13 empty.

5. Enter formulae and values into cells B7 through C11 to define the prey and predator ZNGIs.

This will force the spreadsheet to plot the ZNGIs on the graph, as shown in Figure 1.

In cell B7, enter the formula **=MAX(B14:B114)**.
In cell C7, enter the formula **=$F$7/$H$9**. This corresponds to R/a, the equilibrium value of the prey population (see Equation 3).

Cells B7 and C7 are the coordinates of the right-hand end of the prey ZNGI. Of course, this line extends infinitely to the right, but we cut it off even with the maximum actual value of the prey population so that we can graph our results.

In cell B8, enter the value 0. Copy the formula from cell C7 into cell C8.
Cells B8 and C8 are the coordinates of the point where the prey ZNGI intersects the predator (vertical) axis.

In cells B9 and C9, enter the value 0.
Cells B9 and C9 are the coordinates of the origin of the graph. This is a trick to get us from the prey ZNGI to the predator ZNGI without drawing extraneous lines on the graph.

In cell B10 enter the formula **=$H$7/($H$9*$H$8)**. This corresponds to q/af, the equilibrium value of the predator population (see Equation 4).
In cell C10, enter the value 0.
Cells B10 and C10 are the coordinates of the point where the predator ZNGI intersects the prey (horizontal) axis.

Copy the formula from cell B10 into cell B11.
In cell C11, enter the formula **=MAX(C14:C114)**.
Cells B11 and C11 are the coordinates of the upper end of the predator ZNGI. Like the prey ZNGI, this line is infinitely long, but we truncate it at the maximum predator population for convenience.

6. Enter a formula to calculate the size of the prey population at time 1.

In cell B15, enter the formula **=IF(B14+$F$7*B14-$H$9*C14*B14>0, B14+$F$7*B14-$H$9*C14*B14,0)**.

B14+$F$7*B14-$H$9*C14*B14 corresponds to Equation 1,

$$V_{t+1} = V_t + RV_t - aC_tV_t$$

However, if you simply use Equation 1, it is likely to produce negative population sizes, which make no sense biologically. We use the **IF()** function here to prevent this population from going negative. The formula says, “Calculate the prey population according to Equation 1, and if the result is greater than zero, use it. If the result is zero or less, use zero.”

You can simplify the task of entering this formula if you type it in through the “>0,” copy the part between the left parenthesis and the “>“ sign, and paste it after the comma. Then type in the second comma, followed by a zero, and close the parentheses.

7. Enter a formula to calculate the size of the predator population at time 1.

In cell C15, enter the formula **=IF(C14+$H$8*$H$9*B14*C14$H$7*C14>0, C14+$H$8*$H$9*B14*C14-$H$7*C14,0)**. **C14+$H$8*$H$9*B14*C14-$H$7*C14** corresponds to Equation 2,

$$C_{t+1} + afV_tC_t - qC_t$$

Here again, we use the **IF()** function to prevent the population from going negative. You can use the same shortcut to enter this formula as in the previous step.

8. Copy the formulae from cells B16 and C16 down their columns.

Select cells B15 through C15. Copy.
Select cells B16 through C114. Paste.

9. Save your work.

B. Create graphs.

1. Graph prey and predator populations against time. Edit your graph for readability.

Be aware that the predator population is plotted on a different scale (the right-hand y-axis) than the prey population (the left-hand y-axis). This is necessary because the two cover such different ranges.

Select cells A14 through C114. Follow the usual procedure to make an XY graph. In the second Chart Wizard dialog box, click on the Series tab, and use the boxes to name Series 1 "Prey" and Series 2 "Predator."



Figure 3

After you've finished the graph, double-click on a data point in the line for the predator population. This line will lie almost on top of the x-axis, so it may take several tries to select the data series rather than the axis. In the Format Data Series dialog box, click on the Axis tab, and select Secondary Axis. This will cause the predator population to be plotted on a separate y-axis, with a different scale from that of the prey population. Your graph should resemble Figure 3.

See Exercise 8, "Logistic Population Models," for details on creating a second y-axis.

2. Graph predator population (y-axis) against prey population (x-axis), as in the standard presentation of the Lotka-Volterra model. Edit your graph for readability.

Select cells B7 through C114 and make an XY graph.

In the third Chart Wizard dialog box, click the Legend tab and click in the Show Legend checkbox, to prevent the legend from being shown (the check mark in the box should disappear). Your graph should resemble Figure 4.



Figure 4

You should see that the trajectory spirals in a counterclockwise direction.
Your graph will show the two ZNGIs, but unfortunately will not label their endpoints.

The graph will also not indicate which direction (clockwise or counterclockwise) the population trajectory moves. You can figure this out by locating the point (V_0, C_0), which is the first point on the trajectory.

QUESTIONS

1. Does a larger prey population growth rate (R) increase or decrease the stability of the predator-prey interaction?
2. What happens if the predators starve more quickly? Less quickly?
3. What happens if the predator is more efficient at converting prey into offspring? Less efficient?
4. What happens if the predator is better at finding prey? Worse?
5. Is the behavior of the model sensitive to starting populations? Begin with populations near the point where the isoclines cross, and move slowly farther out.
6. What is the ultimate outcome of the predator-prey interaction, regardless of parameter values? How does this compare to real predator and prey populations? What factors not included in the model may explain the differences between model predictions and reality?

Modifying the Model to Include Prey Refuges

In the model so far, predators are capable of hunting down every single prey individual. In reality, it is often the case that some prey individuals can escape predation by hiding in refuges, such as burrows, crevices in rocks or coral reefs, etc. Thus, there will always be at least a few prey individuals surviving. These survivors, of course, could potentially breed and replenish the prey population. Does the presence of prey refuges alter the outcome of the model?

INSTRUCTIONS	ANNOTATION
Part 2. Predator-Prey Model with Prey Refuges	
A. Set up the spreadsheet.	
1. Return the parameters to their original values (see Figure 2).	If you wish to retain your existing model, save it under a separate file name before making changes, or copy your spreadsheet to a new worksheet and make changes on the copy.
2. Modify your existing formula for the prey population at time 1 to include prey refuges.	Edit the formula in cell B15 by changing the zeros to tens. The new formula should read **=IF(B14+$F$7*B14-$H$9*B14*C14>10,B14+$F$7*B14-$H$9*B14*C14,10)**. This formula says to calculate the size of the prey population at time 1 based on its size at time 0 and losses to predation. If that size is greater than 10, use it; otherwise, make the prey population 10. The biological interpretation is that at least 10 prey individuals survive in refuges, regardless of the number or effectiveness of predators.
3. Copy the modified formula down its column.	Copy the formula in cell B15 into cells B16 through B114.
4. Try other values for the number of survivors.	Repeat steps 2 and 3, using some number other than 10.
B. Create graphs.	You do not need to make any new graphs or edit your existing ones. Your changes will be automatically reflected in your existing graphs.

QUESTIONS

7. Reinvestigate questions 1–6 on the preceding page, but based on your model with prey refuges.

Modifying the Model to Include a Prey Carrying Capacity

The classical continuous-time Lotka-Volterra predator-prey model predicts that prey and predator populations will cycle endlessly around their equilibrium values. Some real predator-prey systems, such as the snowshoe hare and Canada lynx, display cycles that resemble these, but others do not. Even in cases of cyclic population dynamics, ecologists seriously question whether the Lotka-Volterra model, with all its simplifying assumptions, accurately reflects reality. A recent model of the hare-lynx cycle (King and Schaffer 2001) includes 17 parameters and variables.

One obvious omission from the Lotka-Volterra model is any limitation on the prey population other than losses to predation. Surely, prey individuals require resources such as food and water, which could potentially limit the size of their population even in the absence of predators. Perhaps including a prey **carrying capacity** in the model would reduce its tendency to cycle, or in the case of the discrete-time model, its tendency toward increasing population fluctuations and eventual extinctions. In other words, if there were a cap on the size of the prey population, that number might also limit the predator population, which in turn might prevent the predators from hunting the prey to extinction and then starving.

We can modify our prey population equation, Equation 1, to include a carrying capacity in the same way we modified our geometric population equation in Exercise 5, "Logistic Population Models." If we let K_v represent the prey carrying capacity (in the absence of predators), we can write

$$V_{t+1} = V_t + RV_t\left(\frac{K_v - V_t}{K_v}\right) - aC_tV_t \qquad \text{Equation 5}$$

If the predator population (C_t) is zero, then losses to predation (aC_tV_t) will be zero, and the prey population will stabilize at K_v. If predators are present, losses to predation will reduce the prey population to some value less than K_v. We will leave the predator equation unchanged for now.

Equilibrium Solution. Because we have not changed the predator equation, its equilibrium solution remains unchanged. However, our change in the prey equation means we must solve the new equation for its equilibrium (ZNGI). We find this by setting $\Delta V_t = 0$.

$$\Delta V_t = V_{t+1} - V_t = RV_t\left(\frac{K_v - V_t}{K_v}\right) - aC_tV_t$$

$$0 = RV_t\left(\frac{K_v - V_t}{K_v}\right) - aC_tV_t$$

$$aC_tV_t = RV_t\left(\frac{K_v - V_t}{K_v}\right)$$

$$aC_t = R\left(\frac{K_v - V_t}{K_v}\right)$$

$$C_t = \frac{R}{a}\left(1 - \frac{V_t}{K_v}\right)$$

$$C_t = \frac{R}{a} - \frac{RV_t}{aK_v}$$

$$C_t = \frac{R}{a} - \frac{R}{aK_v}V_t$$

There's no easy way to express this equilibrium solution in words, but we can deduce some things about it. First, the equation is in the standard form of a straight line ($y = a + bx$), with a slope of $-R/(aK_v)$. Second, if we plug in $V_t = 0$, we find the y-intercept (C-intercept) to be R/a, just as in the classical Lotka-Volterra model. Third, if we plug in $C_t = 0$, we find the x-intercept (V-intercept) to be K_v (see below). This makes sense, because we would expect the prey population to go to K_v if there were no predators present.

$$0 = \frac{R}{a} - \frac{R}{aK_v}V_t$$

$$\frac{R}{aK_v}V_t = \frac{R}{a}$$

$$\frac{V_t}{K_v} = 1$$

$$V_t = K_v$$

INSTRUCTIONS	ANNOTATION

Part 3. Predator-Prey Model with Prey Carrying Capacity

A. Set up the spreadsheet.

1. Return the parameters to their original values.

To retain your existing model, save it under a separate file name before making changes, or copy your spreadsheet to a new worksheet and make changes on the copy.

2. Modify your existing spreadsheet headings to include a prey carrying capacity.

Edit the text in cell A2 to reflect the change to a logistically-growing prey population.
In cell E8, enter the label "Kv".
In cell F8, enter the value 2000.

3. Enter formulae and values into cells B7 through C11 to define the prey and predator ZNGIs.

Your graphs will look very odd while you are making these changes. Ignore them for now—the errors will disappear after you complete the changes to your spreadsheet.
In cell B7, enter the formula **=$F$8**.
In cell C7, enter the value 0.
Cells B7 and C7 are the coordinates of the point where the prey ZNGI crosses the prey axis, (K_v,0). Leave cells B8 through C11 unchanged.

4. Modify the formula for the prey population at time 1 to include the prey carrying capacity.

In cell B15, enter the formula **=IF(B14+$F$7*B14*($F$8-B14)/$F$8-$H$9*B14*C14>0, B14+$F$7*B14*($F$8-B14)/$F$8-$H$9*B14*C14,0)**.
B14+$F$7*B14*($F$8-B14)/$F$8-$H$9*B14*C14 corresponds to the equation

$$V_{t+1} = V_t + RV_t\left(\frac{K_v - V_t}{K_v}\right) - aC_tV_t$$

which is our logistic model of the prey population. Again, we use the **IF()** function to prevent the population from going negative.
Note that we removed the refuges from the prey population by changing the >10 back to >0. We do this so we can see the effects of a prey carrying capacity without clouding the issue with refuges.

5. Copy the modified formula down its column.

Select cell B15. Copy. Select cells B16 through B115. Paste.
Your spreadsheet should resemble Figure 5.

	A	B	C	D	E	F	G	H
1	***Predator-Prey Dynamics***							
2	Uses a logistically-growing prey population, with an additional term for losses to predators.							
3	Uses an exponentially-growing predator population with per capita pop growth rate determined							
4		by prey capture and conversion efficiency.						
5								
6		**Zero net growth isoclines**			**Prey parameters**		**Predator parameters**	
7		2000.000	0.000		*R*	0.250	Starvation rate (*q*)	0.100
8		0.000	25.000		K_v	2000.000	Conversion efficiency (*f*)	0.008
9		0.000	0.000				Attack rate (*a*)	0.010
10		1250.000	0.000					
11		1250.000	19.600					
12								
13	Time							
14	0	1000.000	20.000					
15	1	925.000	19.600					
16	2	867.997	19.090					

Figure 5

B. Create graphs.

You do not need to make any new graphs. Your existing graphs will automatically reflect the changes in your spreadsheet. Edit the graph titles to distinguish them from graphs of the classical Lotka-Volterra model. Your graphs should now resemble Figures 6 and 7.



Figure 6



Figure 7

QUESTIONS

8. Reinvestigate questions 1–6 but based on your model with a carrying capacity for the prey population.

Modifying the Model to Include Carrying Capacities for Prey and Predator

It is quite conceivable that the predator population may have a carrying capacity imposed by environmental constraints other than prey availability. Factors imposing such a limitation might include mutual interference between predators (fighting over prey or hunting territories) or limited availability of other essential resources, such as water, burrow sites, or something else. If prey are superabundant (i.e., supply exceeds demand and no predators starve), then the predator population (C_t) will increase to its carrying capacity (K_c), but not beyond it.

We can include a predator carrying capacity in the same way we included a prey carrying capacity. We will modify the predator equation as follows:

$$C_{t+1} = C_t + afV_tC_t\left(\frac{K_C - C_t}{K_C}\right) - qC_t$$

Will the introduction of a predator carrying capacity change the behavior of the model? Try predicting the result before exploring it with the spreadsheet.

Equilibrium Solution. As before, we will have to re-derive our equilibrium solution for this modified equation. Letting $\Delta C_t = 0$, we get

$$0 = afV_tC_t\left(\frac{K_c - C_t}{K_c}\right) - qC_t$$

$$qC_t = afV_tC_t\left(\frac{K_c - C_t}{K_c}\right)$$

$$q = afV_t\left(\frac{K_c - C_t}{K_c}\right)$$

$$\frac{q}{af} = V_t\left(\frac{K_c - C_t}{K_c}\right)$$

$$\frac{qK_c}{af(K_c - C_t)} = V_t$$

In words, "Gadzooks!" But it turns out this produces a predator ZNGI that crosses the x-axis (V-axis) at the same point as before, $V = q/af$ (plug in $0 = C_t$ and solve). However, instead of a straight vertical line, it gives us a curve that leans over to the right, as you will see in the spreadsheet graph. The ZNGI equation makes no sense at $C_t = K_c$, because the denominator of the term on the left becomes undefined, and then negative.

INSTRUCTIONS	ANNOTATION
Part 4. Predator-Prey Model with Carrying Capacities for Prey and Predator	
A. Set up the spreadsheet.	If you wish to retain your existing model, save it under a separate file name before making changes, or copy your spreadsheet to a new worksheet and make changes on the copy.
1. Change your parameters to these values: $q = 0.25$, $f = 0.20$, $a = 0.005$	Enter the values given into cells H7, H8, and H9, respectively.
2. Modify your existing spreadsheet headings to include a predator carrying capacity.	Edit the text in cell A3 to reflect the change to a logistically growing predator population. In cell G10, enter the label "Kc". In cell H10, enter the value 100.
3. Change the initial population sizes to $V_0 = 100$, $C_0 = 10$.	Enter the given values into cells B14 and C14, respectively.
4. Enter formulae and values into cells B8 through C12 to define the prey and predator ZNGIs.	Your graphs will look very odd while you are making these changes. Ignore them for now—the errors will disappear after you have completed all the changes to your spreadsheet. Leave cells B8 through C10 unchanged. Delete the contents of cells B11 and C11.
5. Modify the formula for the predator population at time 1 to include the predator carrying capacity.	In cell C15, enter the formula **=IF(C14+$H$8*$H$9*B14*C14*($H$10-C14)/$H$10-$H$7*C14>0,C14+$H$8*$H$9*B14*C14*($H$10-C14)/$H$10-$H$7*C14,0)**. This corresponds to Equation 6: $$C_{t+1} = C_t + afV_tC_t\left(\frac{K_C - C_t}{K_C}\right) - qC_t$$ Again, we use the **IF()** function to prevent the population from going negative.
6. Copy the modified formula down its column.	Select cell C15. Copy. Select cells C16 through B114. Paste.
7. Set up a new data series in column D to graph the predator ZNGI.	We need to do this because this ZNGI is not a straight line, so we must calculate many points along it, and connect them with a line. We will use the formula we derived above to express the predator ZNGI as a function of prey population size: $$C_t = K_c - \frac{qK_c}{afV_t}$$ We must use a little spreadsheet trickery to make this come out right on the graph. Indeed, even with our trickery, the ZNGI may look a little strange with some parameter values.

In cell B13 enter the formula **=$H$7/($H$9*$H$8)**. This is equal to $q/(af)$.
Leave cell C13 empty. In cell D13, enter the value 0.

In cell D14, enter the formula **=IF($H$10-($H$7*$H$10)/($H$9*$H$8*B14)>0,$H$10-($H$7*$H$10)/($H$9*$H$8*B14),0)**.
Use the same shortcut as before to enter this formula.

This formula requires a little explanation. It is the spreadsheet version of the equation for the predator ZNGI (derived above), rewritten as a function of V_t, so that we can plot it on the graph of predator population versus prey population. The derivation is:

$$\frac{qK_c}{af(K_c - C_t)} = V_t$$

$$qK_c = af(K_c - C_t)V_t$$

$$\frac{qK_c}{V_t} = afK_c - afC_t$$

$$afC_t = afK_c - \frac{qK_c}{V_t}$$

$$C_t = \frac{afK_c}{af} - \frac{qK_c}{afV_t}$$

$$C_t = K_c - \frac{qK_c}{afV_t}$$

Copy the formula from cell D14 into cells D15 through D114. Your spreadsheet should look like Figure 8.

	A	B	C	D	E	F	G	H
1	***Predator-Prey Dynamics***							
2	Uses an exponentially-growing prey population, with an additional term for losses to predators.							
3	Uses an exponentially-growing predator population with per capita pop growth rate determined							
4		by prey capture and conversion efficiency.						
5								
6		**Zero net growth isoclines**			**Prey parameters**		**Predator parameters**	
7		2000.000	0.000		*R*	0.250	Starvation rate (*q*)	0.250
8		0.000	50.000		K_v	2000.000	Conversion efficiency (*f*)	0.200
9		0.000	0.000				Attack rate (*a*)	0.005
10		250.000	0.000				K_c	100.000
11								
12								
13	Time	250.000		0.000				
14	0	100.000	10.000	0.000				
15	1	118.750	8.400	0.000				
16	2	141.687	7.214	0.000				

Figure 8

B. Create graphs.

1. Make a new graph of predator population versus prey population, including the new ZNGIs. Edit your graph for readability. It should resemble Figure 9.

It is possible to edit your existing graph, but that is difficult and prone to error, so it's easier just to start over.

Select cells B7 through D114 and make an XY graph.
Select the predator ZNGI by double-clicking on any data point along it. In the Format Data Series dialog box, click the Patterns tab and choose None for marker style. This will cause the predator ZNGI to be plotted as a line with no data markers, like the prey ZNGI.

2. Do not change your graph of population sizes versus time.



Figure 9

QUESTIONS

9. Reinvestigate questions 1–6 but based on your model with carrying capacities for both prey and predator populations.
10. Attempt to summarize the implications of all the models developed in this exercise.

LITERATURE CITED

King, A. A. and W. M. Schaffer. 2001. The geometry of a population cycle: A mechanistic model of snowshoe hare demography. *Ecology* 82: 814–830.

Rosenzweig, M. L. and R. H. MacArthur. 1963. Graphical representation and stability conditions of predator-prey interactions. *American Naturalist* 97: 209–223.